

Asservissement de la puissance émise par une antenne radio

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :


[liste des PDF](#)

L'article précédent (n° 94) a montré un cas théorique d'un asservissement de puissance dans une résistance. On se propose ici de prolonger l'étude sur un cas concret : l'asservissement de puissance émise par une antenne de téléphone portable. Pour se rapprocher au mieux d'un cas industriel, il a été choisi une transmission GSM. Mais pour faciliter le lien avec l'article 94, il a été fait des hypothèses simplificatrices de la partie radio fréquence (RF).

1) Analyse du schéma

1.a Environnement complet

On analyse l'asservissement de puissance de la partie radio fréquence d'un téléphone portable GSM, 2G.



Asservissement de la puissance émise par le téléphone portable vers la base fixe.

Dans ce type de communication, la modulation est de type GMSK, émise 1/8 du temps.

En effet, de façon transparente à l'utilisateur, l'émission n'est pas continue durant une conversation, mais se fait par slot temporel, appelé burst. La puissance émise doit obéir à 2 critères :

- la montée et la descente du burst doit suivre un gabarit (pour des raisons d'encombrement spectral),
- son niveau doit être dimensionné à une valeur la plus faible possible (pour réduire la consommation électrique et le rayonnement dans le corps de l'utilisateur), tout en restant suffisamment élevé pour avoir un taux d'erreur par bit acceptable (certains bits connus sont placés dans la trame transmise ce qui permet de vérifier leur intégrité par la base, en temps réel).

La boucle d'asservissement est analogique.

Nous allons prendre un exemple concret, expliqué plus longuement sous [cours communications numériques](#) en pages 141, 142, 143 et pages 150, 151. On rappelle ci-dessous les principales caractéristiques :

- Le signal GMSK est une onde modulée en fréquence par le flot binaire à transmettre. La fréquence porteuse est placée entre 880 MHz et 915 MHz, soit une période d'environ 1,1 ns,
- le burst dure 577 μ s,
- l'amplificateur radio fréquence est un BGY240,
- l'adaptation d'impédance est réalisée par une structure passe bas avec $L = 3,3$ nH et $C = 1,5$ pF,

- la mesure de puissance se fait par un coupleur unidirectionnel formé de 2 pistes parallèles de 3 cm de long, espacées de 0,2 mm, la piste secondaire faisant 0,2 mm de large, et la piste principale faisant 0,6 mm de large,
- la conversion entre la puissance captée et une tension exploitable est réalisée par un détecteur de crête constitué d'une diode Schottky BA570, d'un condensateur de 27 pF, et d'une résistance 10 kΩ.

La consigne de puissance est un signal analogique, issu d'un circuit qui, d'après une information envoyée par la base, détermine le niveau à appliquer à l'antenne d'émission.


La partie radio fréquence présente de nombreuses inconnues : comportement du BGY240, comportement des micro ruban couplés, comportement du duplexeur, pertes de l'antenne, ce qui rend une modélisation représentative impossible avec Pspice. Nous allons faire de nombreuses hypothèses simplificatrices.

1.b Modélisation du circuit d'antenne en HF

Le modèle électrique d'une antenne est un dipôle, présentant une partie imaginaire, schématisée ci-contre par un circuit LC série, et une partie réelle $R_{ant} = R_p + R_{ray}$.

R_p : résistance symbolisant les pertes (par effet Joule, par onde de surface),

R_{ray} : résistance fictive symbolisant la puissance rayonnée dans l'espace : les watts émis sont ceux absorbés par R_{ray} . Sa valeur est liée à l'environnement, à la propagation des ondes etc.



Modélisation électrique de liaison HF

L'amplificateur haute fréquence BGY240 peut être représenté par sa source de Thévenin : une source parfaite avec, en série, la résistance de sortie, que nous allons négliger.

Le circuit d'adaptation d'impédance est conçu pour annuler la partie imaginaire en travaillant à la résonance (consulter [annexe 1](#)). La maille est alors équivalente à une résistance unique, R_a .

Pour simuler l'asservissement de puissance, on va fixer arbitrairement la puissance débitée par l'amplificateur, que l'on considérera parfait, à 1 W.

Avec une alimentation de 3 V, la tension sinusoïdale en sortie aura 1,5 V d'amplitude, composante continue bloquée, soit $(1,5/\sqrt{2})$ V efficace.

Par loi de Joule, on a $R_a = U^2 / P = 1,125 \Omega$.

Remarque : on peut exploiter la loi de Joule avec un signal sinusoïdal pur ou modulé en fréquence, la valeur efficace de la tension et la puissance restent identiques.





Schéma de travail

1.c Modélisation du BGY240


Le BGY 240 est un amplificateur UHF en module.

La data sheet ne donne que le minimum pour sa constitution interne : un amplificateur à 3 étages, avec une entrée de contrôle, le tout en un boîtier SOT 388 C, avec 4 broches + potentiel du boîtier à souder sur GND.



Intérieur du BGY 240

Il s'agit donc d'un amplificateur à 2 entrées. Pour simuler l'asservissement de puissance, on peut se contenter d'un modèle constitué d'une source de tension parfaite commandée par le produit des 2 entrées, via un coefficient.



Modélisation du BGY 240 proposée : un multiplieur parfait

De cette façon, la tension de sortie est celle d'entrée multipliée par la tension de commande.


Rappelons que l'aspect « hautes fréquences » ne concerne pas l'asservissement de puissance. C'est pourquoi on a adopté une modélisation simplifiée de l'ampli HF et du circuit d'antenne.

1.d Modélisation du coupleur

Le captage de puissance se comporte - très grossièrement - comme un transformateur : une tension secondaire est provoquée par un courant circulant dans le primaire. Pour modéliser cet étage, de façon unidirectionnelle, on peut choisir une source de tension commandée par un courant via un coefficient. L'information sortante est une tension alternative.

Remarque : sous Pspice, pour accéder, en cours de simulation, à un courant dans une branche, il est nécessaire d'y placer une source de tension en série. De choisir une tension nulle ne modifie en rien le fonctionnement.

La conversion en une tension est assurée par une source commandée de type « H ».



1.e Le détecteur de crête

Le montage est aussi appelé détecteur d'enveloppe quand il est utilisé en démodulation d'amplitude. Dans ce cas, ce montage nécessite de respecter la double inégalité : $T_{modulant} \gg RC \gg T_{porteuse}$.




Sur le téléphone GSM, le signal RF a une fréquence proche du GHz. Le coupleur délivre donc une onde de période voisine de 1 ns. Le gabarit de montée et descente en puissance impose une transition complète en 30 µs environ. Il faut donc que l'information issue du détecteur soit bien plus brève, pour ne pas ralentir la transition.

On a alors $30 \mu s \gg \tau = RC \gg 1 ns$.

La cellule du téléphone dispose de $R = 10 k\Omega$ et $C = 27 pF$, soit $RC = 270 ns$. La double inégalité est respectée.


Le comportement à la montée est instantané, alors que la décharge du condensateur C est présente à la descente.



Pour notre simulation, pour des raisons de temps de calcul, et d'affichage des résultats, on va fixer une porteuse RF de 100 MHz au lieu de 1 GHz.

Néanmoins, la cellule $R // C$ peut être conservée : $30 \mu s \gg RC = 270 ns \gg 10 ns$.

De même, on sera amené à simuler un burst plus court, afin de gagner en temps de simulation.




1.f Schéma simulé

Il vient un schéma de simulation, dont certains éléments du téléphone sont remplacés par des sources commandées parfaites :

Le filtre passe bas est un circuit $R_f = 1 k\Omega$ et $C_f = 100 pF$, donc de fréquence de coupure 1,6 MHz.

On remarque l'ajout d'un écrêteur en sortie du comparateur : c'est pour modéliser le fait que la tension de sortie ne peut pas sortir des alimentations.

On appelle Pant, la puissance antenne, soit, d'après nos hypothèses simplificatrices, la puissance délivrée par l'amplificateur RF.



Le téléphone examiné exploite, par le détecteur de crête, l'amplitude (et non l'amplitude au carré) pour disposer du retour de puissance.

2) Simulations annexe 2, fichier `antenne.cir`

2.a Caractéristique de transfert $P_{ant} = f(V_c)$ à amplitude du signal GMSK = 1,5 V

Dans notre schéma, la puissance d'antenne ne peut être obtenue que par une simulation transitoire. Le run dure 1 μ s, durée largement suffisante pour que les courbes soient stabilisées.

La puissance dégagée par l'antenne est, par nos hypothèses de modélisation, celle dissipée par loi de Joule dans Ra. Elle est disponible par le menu Probe par la fonction pré établie W() pour avoir la puissance instantanée et AVG(W()) pour avoir sa valeur efficace par intégration. On place le curseur à 0,9 μ s.


X Values	900.000n
AVGX(W(Ra),50n)	31.914m
AVGX(W(Ra),50n)	364.256m
AVGX(W(Ra),50n)	1.0240
AVGX(W(Ra),50n)	1.9834
AVGX(W(Ra),50n)	3.2413

Puissance antenne pour 0,1 V 0,5 V 1 V 1,5 V 2 V de la tension de consigne

En réalisant 20 simulations pour 20 tensions de consigne (par la directive .STEP), on obtient 20 courbes.

Un report des coordonnées du curseur dans Excel permet d'afficher la courbe de puissance en fonction de la tension de consigne. Elle est de type quadratique.


On remarque que la boucle est réglée de façon à avoir quasiment 1 W pour 1 V en consigne de puissance.



Puissance antenne fonction de la tension de consigne

2.b Réponse à l'échelon

Pour tester notre asservissement en dynamique, on va placer un signal de consigne de puissance qui change instantanément de valeur de 0 V à 1 V (montée et descente en 1 ns). Pour gagner en temps de simulation, on propose de limiter la durée de la tension de consigne à 2 μ s.



Tension de consigne de puissance : échelon (niveau haut = 1 V durant 2 μ s) et puissance dissipée correspondante

Interprétation :

La consigne passe à 1 V. La puissance de sortie est alors très proche de 1 W, déjà mesuré précédemment. La puissance s'établit et s'arrête en moins de 0,2 μ s.

En conclusion, ce montage présente des performances dynamiques suffisantes pour installer / désinstaller une puissance de 1 W en suivant un profil qui dure 30 μ s.

2.c Exploitation : profil de puissance

La montée et descente de la puissance doivent entrer dans un gabarit afin de respecter des normes GSM.

Usuellement, on présente la puissance sur une échelle log, en exploitant : $P \text{ (dBW)} = 10 \log \left(\frac{P}{1 \text{ W}} \right)$.

On a ainsi : 1 W qui représente 0 dBW, 1 mW qui représente -30 dBW.

On exploite également, pour de faibles puissances, $P \text{ (dBmW)} = 10 \log \left(\frac{P}{1 \text{ mW}} \right)$.

On a ainsi : 1 mW qui représente 0 dBmW, 1 μW qui représente -30 dBmW.

En GSM, la dynamique de puissance va de 5 dBmW (soit 3,2 mW) à 33 dBmW (soit 2 W), par paliers de 2 dBmW (soit 1,259 mW). Il y a donc 14 valeurs possibles de puissance (réparties régulièrement sur une échelle log), et donc 14 valeurs possibles sur la tension de commande.


Dans notre simulation, pour une facilité de lecture, nous allons placer 4 valeurs de tension de commande.

On propose un gabarit donné par les coordonnées de 2 premières colonnes suivantes :


temps	Puissance émise	Vc nécessaire
3 μs	0,25 W	0,3672
9 μs	0,5 W	0,606
18 μs	0,75 W	0,8135
30 μs	1 W	0,9922

Par la fonction Vc = f (Pant), on détermine les valeurs que doit générer Vc, valeurs placées dans la troisième colonne.

Un léger ajustement des temps et/ou des valeurs numériques de tensions de consigne permet de compenser le temps de réponse identifiés précédemment.



exemple de montée en puissance
1 graduation = 3 μs




Montée et descente de puissance du burst, sur une échelle linéaire. 1 graduation = 3 μs.
Ici, pour une meilleure visualisation des transitions, le burst est limité à 15 μs, (au lieu de 577 μs).

Interprétation

- Pour une meilleure visibilité, l'ordre de montée en puissance commence à t = 6 μs, soit 2 graduations.
- Les montées et descente de la puissance émise par l'antenne respectent la loi de commande imposée par la tension de consigne.
- La puissance « de croisière » (le palier haut) est bien 1 W.

On peut également vérifier quelques potentiels stratégiques de notre montage. On choisit le régime établi, vers 45 μ s, c'est à dire à Pa \approx 1 W.



A la puissance nominale de 1 W dans Ra, courant, signal de retour et tension de commande du BGY240

Interprétation


Le courant circulant dans Ra est sinusoïdal, d'amplitude 1,331 A. Sa valeur efficace est alors $1,331 \text{ A} / \sqrt{2} = 0,941 \text{ A}$. (On retrouve ainsi la puissance par Ra $I^2 = 1,125 \times 0,941^2 = 0,996 \text{ W}$).

En V(32), une image de cette valeur de courant est 430 mV. Cette valeur s'explique par la conversion I/U faite dans la source H, et du comportement de la diode Schottky.

En raisonnant par le fonctionnement d'un système bouclé, *mais sur les tensions*, on obtient une erreur statique $V(4) - V(32) = 1 - 0,43 = 0,57 \text{ V}$.

2.d Robustesse

La résistance de rayonnement est très liée à l'environnement. Ce dernier est fluctuant selon le déplacement, l'orientation du téléphone etc. Pour vérifier la robustesse de l'asservissement, on renouvelle des 2 dernières simulations à $R_a = 1,125 / 2 = 0,5625 \Omega$ et à $R_a = 2 \times 1,125 = 2,25 \Omega$.



Interprétation

On a sensiblement le même fonctionnement : à $R_a = 0,5625 \Omega$ la puissance est de 840 mW, et pour $2,25 \Omega$, la puissance est de 970 mW.

CONCLUSION

Cet article a montré un asservissement de puissance concret. Malgré les hypothèses simplificatrices utilisées (la partie radio fréquence vue de façon simplifiée), les courbes obtenues après simulations PSpice sont parfaitement réalistes.


Annexe 1: fichier **adaptation.cir**

Vérification de l'accord à la fréquence centrale 902 MHz du circuit d'adaptation de l'antenne.

Remarque : $902 = \sqrt{(890 \times 915)}$.

Le modèle de l'antenne (en très hautes fréquences) est un circuit LC Ra.

La simulation ci-dessous montre qu'avec $L = 3,85 \text{ nH}$, $C = 4 \text{ pF}$, $R_a = 1,125 \Omega$, la partie imaginaire est nulle à 902 MHz.



adaptation par LC

* fichier adaptation.cir

Vgmsk 1 0 AC=1

L5540 1 2 3.3nH

C560 2 0 1.5pF

Lant 2 3 3.85n

Cant 3 4 4pF

.param Ra=1.125

Ra 4 0 {Ra}


.AC DEC 1000 100MEG 10G

.step param Ra list 0.5625 1.125 2.25

.probe

.end

Netliste, prête à simuler



Interprétation

On teste à $R_a = 0,5625 \Omega$, $1,125 \Omega$ et $2,25 \Omega$. On ne présente que [890MHz ; 915 MHz].

La modification de R_a change le facteur de qualité, mais dans l'étendue de la bande, le déphasage reste mesuré.

Annexe 2 : fichier `antenne.cir`

```
asservissement de puissance dans une antenne
* fichier antenne.cir

.MODEL BAS70W D (IS = 3.22E-09 N = 1.018 BV = 77 IBV = 1.67E-07 RS = 20.89
+ CJO = 1.655E-12 VJ = 0.349 M = 0.3583 FC = 0.5 TT = 0 EG = 0.69 XTI = 2)

Vgmsk 1 0 sin (0 1.5 100Meg)
EBGY 3 0 POLY(2) (1,0) (C,0) 0 0 0 0 1 ; 0.6 ; delivre V(1) x V(C) x 0,6

Vnul 30 0 DC=0 ; pour prélever le courant
HU 31 0 POLY(1) Vnul 0 0.6 ; capteur tension commandé par un courant
* 0,6 V pour 1 A

.param Ra = 0.5625 ; 2.25 ; 1.125
Ra 3 30 {Ra} ; 1.125 ;2.25 ; 0.5625 ; modele antenne
D 31 32 BAS70W ;
C 32 0 27p ;
R 32 0 10k ; detecteur de crete

EC 2 0 table {1.8 *(V(4) - V(32))} (0 0 3 3) ;et ampli A
Rf 2 C 1k ; 0,1 us
Cf C 0 100p ; pour la commande du BGY240
*-----
* pour carac Pant = f (consigne)
.param consigne=1
*Vc 4 0 DC={consigne}
*.step param consigne list 0.1 0.5 1 1.5 2
*.TRAN 0.1n 1u 0 0.1n
*-----
* pour mesurer temps de réponse
Vc 4 0 PULSE (0 1 1u 1n 1n 2u 5u )
.TRAN 0.1n 6u 0u 0.1n
*-----
* pour valider gabarit montee et descente
*Vc 4 0 PWL (0us,0V 5.5us,0 9us,0.37V 14.8us,0.62V 24us,0.815V 36us,0.985V
*+ 40us,0.985V 51us,0.985V 63us,0.82V 72us,0.62V 78us,0.37V 81us,0V)
*.TRAN 0.1n 90u 0u 0.1n
.STEP param Ra list 0.5625 1.125 2.25

.OP
.probe
.end
```

Netliste, prête à simuler

Commentaire : par validation/dévalidation des commentaires, on réalise les 3 runs de ce document

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)